Kurvenschar-Diskussion

Analysis mit MuPAD 3.11, mit Kurven der Extrema und der Wendepunkte

Prof. Dr. Dörte Haftendorn, MuPAD 4, http://haftendorn.uni-lueneburg.de Aug.06

Automatische Übersetzung aus MuPAD 3.11, Okt. 05 Version vom Jul.06 update Aug06

Es fehlen noch textliche Änderungen, die MuPAD 4 direkt berücksichtigen, das ist in Arbeit.

Web: http://haftendorn.uni-lueneburg.de www.mathematik-verstehen.de

++

```
f:=(x,k)->x^2*(x-k)*(x-2)
```

$$(x, k) \rightarrow x^2 \cdot (x - k) \cdot (x - 2)$$

```
fani:=plot::Function2d(f(x,k),x=-2..3,k=-1..3,
        ViewingBoxYRange=-3..3, Color=RGB::Red, LineWidth=1):
plot(fani);
```



animieren durch Anklicken!

Die Animation geht auch mit dem einfachen plotfunc2d. Hier soll aber später die Kurve der Extrema hinzugefügt werden. Daher braucht man Graphik-Primitive, wie man sie durch
plot::xxxxxxx erhält. Die werden dann von plot(...) dargestellt.

```
fanika:=plot::Function2d(f(x,k),x=0..2,k=-1..3,
        ViewingBoxYRange=-3..3, Color=RGB::Red, LineWidth=1):
plot(fanika);
```



animieren durch Anklicken!

```
alle:=f(x,k) $ k=-1..3
```

$x^2 \cdot (x+1) \cdot (x-2),\ x^3 \cdot (x-2),\ x^2 \cdot (x-1) \cdot (x-2),\ x^2 \cdot (x-2)^2,\ x^2 \cdot (x-2) \cdot (x$

```
plotfunc2d(alle,x=-1.2..3,YRange=-3..3,
           GridVisible=TRUE)
```



Die Nullstellen sind hier offensichtlich: Gemeinsame doppelte Nullstelle x=0, Berührung,
Gemeinsame Nullestelle bei x=2, weitere Nullstalle bei x=k.
Für k=2 ist auch diese Nullstelle doppelt, also Berührung, sonst ist sie einfach.
für k=0 ist die bei x=0 dreifach, also Sattel, sonst sind alle anderen Nullstellen einfach.
Automatische Bestimmung der Nullstellen:


```
nst=solve(f(x,k)=0,x)
```

$nst = \{0,\ 2,\ k\}$

Die Vielfachheiten sind nicht zu sehen. Die Klammerform ist aussagekräftiger.

```
solve(f(x,k)=0,x, Multiple)
```

$\{[0,\ 2],\ [2,\ 1],\ [k,\ 1]\}$

Multiple gibt die Vielfachheit mit aus.

```
ausmulti:=expand(f(x,k))
```

$2 \cdot k \cdot x^2 - k \cdot x^3 - 2 \cdot x^3 + x^4$

Wenn nun diese Form gegeben wäre, könnte man die Klammerform herstellen.

```
factor(ausmulti)
```

$-x^2 \cdot (x-2) \cdot (k-x)$

Ableitung und zugehörige Rechnungen

```
f
```

$(x,\ k) \rightarrow x^2 \cdot (x-k) \cdot (x-2)$

```
ableit:=diff(f(x,k),x)
```

$x^2 \cdot (x-2) - x^2 \cdot (k-x) - 2 \cdot x \cdot (k-x) \cdot (x-2)$

```
simplify(ableit)
```

$x \cdot (4 \cdot k - 6 \cdot x + 4 \cdot x^2 - 3 \cdot k \cdot x)$

```
ableit:=factor(%)
```

$x \cdot (4 \cdot k - 6 \cdot x + 4 \cdot x^2 - 3 \cdot k \cdot x)$

```
plotfunc2d(subs(ableit, k=2),x=-3..3,YRange=-3..3)
```



```
plotfunc2d(subs(ableit),x=-3..3,k=-2..2,YRange=-3..3)
```




```
subs(ableit, k=2)
```

$$x \cdot (-12 \cdot x + 4 \cdot x^2 + 8)$$

```
ableit
```

$$x \cdot (4 \cdot k - 6 \cdot x + 4 \cdot x^2 - 3 \cdot k \cdot x)$$

```
xe:=solve(ableit=0,x)
```

$$\left\{0, \frac{3 \cdot k}{8} - \frac{\sqrt{9 \cdot k^2 - 28 \cdot k + 36}}{8} + \frac{3}{4}, \frac{3 \cdot k}{8} + \frac{\sqrt{9 \cdot k^2 - 28 \cdot k + 36}}{8} + \frac{3}{4}\right\}$$

```
extremstellen:=float(xe) $ k=-1..2
```

$$\{-0.6930004682, 0.0, 1.443000468\}, \{0.0, 1.5\}, \{0.0, 0.6096117968, 1.640388$$

Achtung, bei Mengen wird die Reihenfolge nicht beibehalten.
ExtremPunkte für k=2:

```
[xex,f(xex,2)] $ xex in subs(xe,k=2);
```

$$[0, 0], [1, 1], [2, 0]$$

```
f(x,2)
```

$$x^2 \cdot (x-2)^2$$

Diese Schar-Funktion hat also, wie man gleich sieht, zwei doppelte Nullstellen und in der Mitte dazwischen ein Maximum.

```
[xex,f(xex,i)] $ xex in subs(xe,k=i) $ i=-1..3:
```



```
matrix([map(%,float)]);
```

$$\begin{pmatrix} 0 & 0 \\ -0.6930004682 & -0.397046341 \\ 1.443000468 & -2.833422409 \\ 0 & 0 \\ 1.5 & -1.6875 \\ 0 & 0 \\ 0.6096117968 & 0.2017155994 \\ 1.640388203 & -0.6196843494 \\ 0 & 0 \\ 1.0 & 1.0 \\ 2.0 & 0 \\ 0 & 0 \\ 1.156929669 & 2.079790697 \\ 2.593070331 & -1.622759447 \end{pmatrix}$$

```
solve(9*k^2-28*k+36=0,k)
```

$$\left\{\frac{14}{9}-\frac{\sqrt{2}\cdot 8\cdot i}{9}, \frac{8\cdot i\cdot \sqrt{2}}{9}+\frac{14}{9}\right\}$$

Die Diskriminate wird nie Null, damit existieren i.a. drei Extrempounkte.

Kurve der Extrema

```
lo:=solve(ableit=0,k)
```

$$\begin{cases} \emptyset & \text{if } x=\frac{4}{3} \\ \mathbb{C} & \text{if } x=0 \\ \left\{-\frac{6\cdot x-4\cdot x^2}{3\cdot x-4}\right\} & \text{if } x\neq 0 \wedge x\neq \frac{4}{3} \end{cases}$$

Die Gleichung Ableitung=0 ist nach k aufzulösen.
Von diesen ist die mittlere die gesuchte.

```
kk:=lo[3][1]
```

$$-\frac{6\cdot x-4\cdot x^2}{3\cdot x-4}$$

Dieser Term ist für k in f(x,k) einzusetzen

```
kurveextrema:=subs(f(x,k),k=kk)
```

$$x^2\cdot (x-2)\cdot \left(x+\frac{6\cdot x-4\cdot x^2}{3\cdot x-4}\right)$$

```
factor(kurveextrema)
```

$$-\frac{x^3\cdot (x-2)^2}{3\cdot x-4}$$

Dieses ist der Funktionsterm der Kurve der Extrema.
Man sieht einen Sattel bei 0, einen Pol bei x=4/3 und eine doppelte Nullstelle bei 2.

```
plotfunc2d(alle,kurveextrema,x=-1.2..3,YRange=-3..3,
GridVisible=TRUE, LineWidth=0.8)
```



Hier sieht man, dass die Kurve der Extrem (hier blau) gut passt.

Wendepunkte

```
zw_abl:=diff(ableit,x)
```

$$4 \cdot k - 6 \cdot x + 4 \cdot x^2 - x \cdot (3 \cdot k - 8 \cdot x + 6) - 3 \cdot k \cdot x$$

```
simplify(zw_abl)
```

$$4 \cdot k - 12 \cdot x + 12 \cdot x^2 - 6 \cdot k \cdot x$$

```
solve(zw_abl=0,x)
```

$$\left\{ \frac{k}{4} - \frac{\sqrt{3} \cdot \sqrt{3 \cdot k^2 - 4 \cdot k + 12}}{12} + \frac{1}{2}, \frac{k}{4} + \frac{\sqrt{3} \cdot \sqrt{3 \cdot k^2 - 4 \cdot k + 12}}{12} + \frac{1}{2} \right\}$$

Das sind unangenehme Terme.

```
solve(3*k^2-4*k+12=0,k)
```

$$\left\{ \frac{2}{3} - \frac{\sqrt{2} \cdot 4 \cdot i}{3}, \frac{4 \cdot i \cdot \sqrt{2}}{3} + \frac{2}{3} \right\}$$

Die Diskriminante wird aber nie Null, damit exitieren stets zwei Wendepunkte.

```
wp:=solve(zw_abl=0,k)
```

$$\begin{cases} \emptyset & \text{if } x = \frac{2}{3} \\ \left\{-\frac{6\cdot x - 6\cdot x^2}{3\cdot x - 2}\right\} & \text{if } x \neq \frac{2}{3} \end{cases}$$

```
wwp:=wp[2][1]
```

$$-\frac{6\cdot x - 6\cdot x^2}{3\cdot x - 2}$$

```
kurvewend:=subs(f(x,k),k=wwp)
```

$$x^2\cdot(x-2)\cdot\left(x+\frac{6\cdot x-6\cdot x^2}{3\cdot x-2}\right)$$

```
factor(kurvewend)
```

$$-\frac{x^3\cdot(x-2)\cdot(3\cdot x-4)}{3\cdot x-2}$$

Auch die Kurve der Wendepunkte hat einen Sattel im Ursprung.
Sie hat einen Pol bei x=2/3 und einfache Nullstellen bei x=4/3 und x=2.
Bemerkenswert ist, dass das Intervall [0,2] von diesen beiden Polstellen genau gedrittelt wird.

```
plotfunc2d(alle,kurveextrema,kurvewend,x=-1.2..3,YRange=-3..
GridVisible=TRUE, LineWidth=0.8)
```



Auch die Kurve der Wendpunkte (hier lila) passt gut.
Nun also alles im animierten Graphen:

```
kuexg:=plot::Function2d(kurveextrema,x=-2..3, LineWidth=1,Co
kuwendg:=plot::Function2d(kurvewend,x=-2..3, LineWidth=1,Col
plot(fani,kuexg,kuwendg)
```



animieren durch Anklicken !!

Integrale

```
int(f(x,k), x)
```

$$\frac{x^5}{5}+\left(-\frac{k}{4}-\frac{1}{2}\right)\cdot x^4+\frac{2\cdot k\cdot x^3}{3}$$

```
i2:=int(f(x,k), x=0..2)
```

$$\frac{4\cdot k}{3}-\frac{8}{5}$$

```
i2 $ k=-1..3
```

$$-\frac{44}{15}, -\frac{8}{5}, -\frac{4}{15}, \frac{16}{15}, \frac{12}{5}$$

Hierzu ist mir nichts Bemerkenswertes eingefallen.

Für welches k wird das Intergral im Intervall [0,2] Null?

```
solve(i2=0,k)
```

$$\left\{\frac{6}{5}\right\}$$

```
in1:=plot::Integral(fanika,10)
```

$$\text{plot::Integral}(\text{Function2d}(-x^2\cdot(k-x)\cdot(x-2),\ x=0..2))$$

```
plot(in1,fani,fanika)
```



animieren durch Anklicken!

```
f(x,6/5)
```

$$x^2 \cdot (x-2) \cdot \left(x - \frac{6}{5}\right)$$

```
plotfunc2d(f(x,6/5), x=-1..3,YRange=-1..1,
        Scaling=Constrained)
```



Bei der Scharkurve mit x=6/5 halten sich also die Flächen über und unter der x-Achse gerade die Waage.

Es folgen einige 3-d-Graphen, die in der Vorlesung entstanden sind.

```
f(x,k)
```

$$-x^2 \cdot (k-x) \cdot (x-2)$$


```
plotfunc3d(f(x,k))
```




```
fxkg:=plot::Function3d(f(x,k),x=-2..3,k=-1..3)
```

$$plot::Function3d(-x^2 \cdot (k-x) \cdot (x-2),\ x = -2..3,\ k = -1..3)$$

```
plot(fxkg)
```




```
fxkg2:=plot::Function3d(f(x,k),x=-2..3,k=-1..3,ViewingBoxZRa
plot(fxkg2)
```




```
g:=(x,y)->x^2*y
```

$(x, y) \rightarrow x^2 \cdot y$

```
form:=plot::Function3d(g(x,y),x=-3..3,y=-1..2): plot(form)
```



Hier wierd nun gezeigt, wie man die Grids wandern lässt.

```
geradeng:=plot::Curve3d([x,y,g(x,y)], x=-2..3,y=-1..3):
parg:=plot::Curve3d([x,y,g(x,y)], y=-1..3,x=-2..3):
 plot(geradeng,parg,form)
```



animieren durch Anklicken!

```
gerg:=plot::Curve3d([x,k,f(x,k)], k=-1..3,x=-1..3,
LineWidth=1, LineColor=RGB::Green):
 plot(gerg,fxkg2)
```



animieren durch Anklicken!

```
polyg:=plot::Curve3d([x,k,f(x,k)], x=-2..3,k=-1..3,
LineWidth=1,LineColor=[1,1,0]):
 plot(polyg,fxkg2)
```



animieren durch Anklicken!